

# COMPOSERS' DESKTOP PROJECT

## CSOUND TUTORIAL INSTRUMENT LIBRARY

**Version 1.0 Richard Orton  
Copyright CDP York, January 1988**

This series of CSound Tutorial Instruments has been devised in order to assist the novice CSound user to get started. The models are those which might be adopted from an understanding of the way an analogue electronic music studio works. This first set of twelve begins with straightforward modules, which are combined in more complex ways as the series proceeds. Each model comprises three elements: the ORCHESTRA file, complete with header; a FLOWCHART showing the design of the instrument as successive modules, and an example SCORE file.

The twelve introduced here are as follows:

- CSTIL01: White Noise Generator with Envelope
- CSTIL02: White Noise Generator with Envelope and Low-Pass Filter
- CSTIL03: Oscillator with envelope
- CSTIL04: Oscillator with pulse envelope and reverberation
- CSTIL05: Sub-Audio Amplitude Modulation
- CSTIL06: Audio-rate Amplitude Modulation
- CSTIL07: Linear and Exponential Glissando Instrument
- CSTIL08: Sub-Audio Frequency Modulation
- CSTIL09: Audio-rate Frequency Modulation
- CSTIL10: Buzz Generator with envelope and controllable number of partials
- CSTIL11: Timbre-migrating instrument
- CSTIL12: Defined pitch-space filled with glissando textures

The way to use these is to begin by trying them as they are, then running them with a new score; and finally try adding further modules of your own to extend the capabilities of the instrument.

### A NOTE ON THE ORCHESTRA HEADER

CSound requires that its control rate (kr) and number of samples per control period (ksmps) must be exact factors of the sample rate. The CDP SoundStreamer interface to Sony's PCM technology demands a sample rate of either 44100 or 22050 per second. With these values there are relatively few factors available for kr and ksmmps:

$$\begin{aligned} 22050 &= 2 \times 3 \times 3 \times 5 \times 5 \times 7 \times 7 \\ 44100 &= 2 \times 2 \times 3 \times 3 \times 5 \times 5 \times 7 \times 7 \end{aligned}$$

The recommended orchestral headers for the two sample rates are:

sr = 22050	sr = 44100
kr = 1470	kr = 1470
ksmps = 15	ksmps = 30
nchnls = 1	nchnls = 1

In the CDP implementation of CSound, if the header is omitted, the lower sample rate with the header as given above is used. If the resolution of the control rate is not enough (e.g. a "clicky" sound results), then changing the control-rate to equal the sample rate, and setting ksmmps = 1, must correct the problem. It will, however, take somewhat longer to create the soundfile.

**;CSTIL01.ORC – White Noise Generator with Envelope****;Uses 7 Score Parameters as follows:****;p1 – always the Instrument Number****;p2 – always the start time in beats (default one second), of the event****;p3 – Duration in beats (default one second), of the event (eg 2)****;p4 – Amplitude (eg 15000)****;p5 – Not Used (no pitch parameter, therefore 0 is placed here)****;p6 – Attack time in secs. for linear envelope (eg .2)****;p7 – Decay time in secs. for linear envelope (eg .3)**

sr = 22050

kr = 1470

ksmps = 15

nchnls = 1


instr 1

k1 linen p4, p6, p3, p7

a1 rand k1

out a1

endin

*Flow Chart for CSTIL01***;CSTIL01.SC – A short example score for a basic noise instrument****;Pfields**

1	2	3	4	5	6	7
Ins.	Start	Dur	Amp	Not Used	Rise Time	Decay Time
No.	Time					
i1	0	2	15000	0	.2	.3
i1	2.2	.2	10000	0	.05	.08
i1	2.5	.2	20000	0	.1	.1
i1	2.75	.3	30000	0	.15	.12
i1	3.25	2.25	20000	0	.05	2.00

e

**;CSTIL02.ORC – White Noise Generator with Envelope and Low-Pass Filter**

;Uses 11 Score Parameters as follows:

;p3 – Duration (eg 2)  
 ;p4 – Amplitude (eg 15000)  
 ;p5 – Not Used (therefore 0 is placed here)  
 ;p6 – Attack time in secs. for linear envelope (eg .2)  
 ;p7 – Decay time in secs. for linear envelope (eg .3)  
 ;p8 – Pch notation of filter frequency setting at start of note  
 ;p9 – Pch notation of filter frequency setting at end of note  
 ;p10 – Rate of change of filter setting  
 ;p11 – Function used to trace filter-setting changes  
 ;(f1 as given here is a straight-line transition, f2 is up and down).

sr = 22050

kr = 1470

ksmps = 15

nchnls = 1

instr 1

p8 = octpch(p8)  
 p9 = octpch(p9)

k1 linen p4, p6, p3, p7


a1 rand k1

k1 oscil p9–p8, p10, p11

a1 tone a1, cpsoct(p8+k1)

out a1

endin



*Flow Chart for CSTIL02*

**;CSTIL02.SC**

;Example Score

;function table one creates a straight line from 0 to 1 over 64 points

f1 0 65 7 0 64 1

;function table two moves from 0 to 1 and back again

f2 0 129 7 0 64 1 64 0

;Fields

1	2	3	4	5	6	7	8	9	10	11
Ins.	Start	Dur	Amp	Not Used	Rise Time	Decay Time	Filter Pch 1	Filter Pch 2	F/Rate (secs)	Function Table No.
No.	Time									
<b>;Noise sweep up</b>										
i1	0	2	15000	0	.2	.3	4.00	10.00	.5	1
<b>;Reverse direction</b>										
i1	2	3	15000	0	.1	.4	9.00	5.00	1.0	1
<b>;Using f2: a bi-directional function</b>										
i1	5	2	20000	0	.5	.3	5.06	8.11	.8	2
i1	7	3	25000	0	.25	.25	9.04	4.09	1.2	2
e										

**:CSTILO3.ORC – Oscillator with envelope**

;The waveform of the oscillator may be selected through p8 pointing to ;wavetable functions 1–3. More may, of course, be added.

;Uses 8 parameters

;p3 – Duration (eg 2)  
 ;p4 – Amplitude (eg 15000)  
 ;p5 – Pitch (eg Pch notation 8.00 for middle C)  
 ;p6 – Attack time of linear envelope (eg .2)  
 ;p7 – Decay time of linear envelope (eg .3)  
 ;p8 – Waveform function select (eg 1, 2 or 3)

sr = 22050

kr = 1470

ksmps = 15

nchnls = 1

instr 1

k1 linen p4, p6, p3, p7  
 a1 oscili k1, cpspch(p5), p8  
 out a1  
 endin

**:CSTILO3.SC**

;Three notes each with a different timbre, resulting from selecting ;a different waveform.

;Arbitrary waveforms

f1 0 512 10 1  
 f2 0 512 10 1 1 1  
 f3 0 1024 10 1 0 1 0 1 0 1 0 1 0 1

;PFields:

No.	Start	Dur	Amp	Pch	Rise	Decay	Function
i1	0	2	15000	8.00	.2	.5	1
i1	2	2	15000	8.01	.2	.5	2
i1	4	2	15000	8.02	.2	.5	3
f0		6.5					


s

;Mixed crazy

t0 300 10 80

i1	0	1	10000	6.00	.02	.05	2
i1	1	1	10000	6.03	.02	.05	3
i1	2	1	10000	7.02	.02	.05	1
i1	3	1	10000	7.06	.02	.05	1
i1	4	1	10000	8.03	.02	.05	3
i1	5	1	10000	9.11	.02	.05	2
i1	6	1	10000	10.05	.02	.05	1
i1	7	1	10000	9.10	.02	.05	3
i1	8	1	10000	8.01	.02	.05	1
i1	9	1	10000	7.04	.02	.05	2
i1	0	1	10000	6.03	.02	.05	2
i1	1	1	10000	6.05	.02	.05	3
i1	2	1	10000	7.07	.02	.05	1
i1	3	1	10000	7.09	.02	.05	2
i1	4	1	10000	8.05	.02	.05	3
i1	5	1	10000	9.14	.02	.05	3
i1	6	1	10000	10.09	.02	.05	2
i1	7	1	10000	9.11	.02	.05	1
i1	8	1	10000	8.03	.02	.05	2
i1	9	1	10000	7.07	.02	.05	1
i1	10	3	10000	7.02	.1	.1	3
i1	10	3	10000	8.04	.1	.1	2
i1	10	3	10000	9.07	.1	.1	1

e



Flow Chart for CSTILO3

**:CSTIL04.ORC – Oscillator with pulse envelope and reverberation**

;Uses 10 parameters  
;:p3 – Duration (eg 2)  
;:p4 – Amplitude (eg 15000)  
;:p5 – Pitch (eg Pch notation 8.00 for middle C)  
;:p6 – Short Attack time of line segment envelope (eg .05)  
;:p7 – Decay time of line segment envelope (eg .3)  
;:p8 – Short Sustain Time (eg .1)  
;:p9 – Waveform function select  
;:p10 – Percentage of sound to be reverberated (eg 40%)

sr = 22050

kr = 1470

ksmps = 15


nchnls = 1

instr 1

i1 = p10\*.01 ;proportion of reverberated sound  
i2 = 1.00 - i1 ;proportion of straight sound

k1 linseg 0, p6, p4, p8, p4, p7, 0, p3-(p6+p7+p8), 0

a1 oscili k1, cspch(p5), p9  
a2 init 0  
a2 reverb a1\*i1, p3  
a1 = a1\*i2  
out a1+a2  
endin



*Flow Chart for CSTIL04*

**:CSTIL04.SC**

;Arbitrary timbres

f1 0 512 10 27 9 3 1 .3 .1


f2 0 512 10 1 .1 .3 .4 .5 .6 .7

f3 0 512 10 1 0 .6 0 .4 0 .2 0 .7

**:PFields**

1	2	3	4	5	6	7	8	9	10
No.	Start	Dur	Amp	Pch	Rise	Decay	Sust.n	Waveform	% Reverb
i1	0	3	18000	8.06	.1	.2	.3	2	50
i1	3	3	15000	6.07	.04	.3	.15	3	60
i1	6	4	20000	8.04	.05	.1	.2	1	40
e									

*linseg produces a pulse envelope, typically:*



**;CSTIL05.ORC Sub-Audio Amplitude Modulation**  
**;Uses 11 parameters**  
**;p3 – Duration**  
**;p4 – Amplitude (eg 15000)**  
**;p5 – Pitch (eg Pch notation 8.00 for middle C)**  
**;p6 – Attack time of linear envelope (eg .2)**  
**;p7 – Decay time of linear envelope (eg .3)**  
**;p8 – Percentage of modulation (eg 40%)**  
**;p9 – First modulation rate in Hz (eg 3)**  
**;p10 – Second modulation rate (eg 5)**  
**;p11 – Function describing transition between p9 & p10 rates (eg 1)**

sr = 22050

kr = 1470

ksmps = 15

nchnls = 1

instr 1

```
k1 linen p4, p6, p3, p7          ;amplitude envelope
k2 = k1 * (p8*.01)              ;amount to be modulated
k1 = k1 - k2                    ;remainder of volume
k3 oscili p10-p9, 1/p3, p11    ;varying am speed over one event
k4 oscili k2, k3+p9, 4          ;am oscil
a1 oscili k4+k1,cpspch(p5),10  ;audio oscil
out a1
endin
```

### **;CSTIL05.SC**

f1 0 65 7 0 64 1


f4 0 129 7 1 64 0 64 1

f10 0 1024 10 1

i1	0	2	15000	8.00	.2	.5	40	3	5	1
i1	2	2	18000	8.07	.3	.4	50	4	7	1
i1	4	2	20000	8.04	.15	.3	60	5	8	1
i1	6	3	24000	9.00	.2	.5	40	3	5	1

e

*Flow Chart for CSTIL05:*



**:CSTIL06.ORC – Audio-rate Amplitude Modulation**

;Uses 11 parameters  
 ;p3 – Duration (eg 2)  
 ;p4 – Amplitude (eg 15000)  
 ;p5 – Pitch (eg Pch notation 8.00 for middle C)  
 ;p6 – Attack time of linear envelope (eg .2)  
 ;p7 – Decay time of linear envelope (eg .3)  
 ;p8 – Modulation Multiplier (eg 1.487)  
 ;p9 – Low Modulation set limit (eg 20Hz)  
 ;p10 – High Mod. set limit (eg 80Hz)  
 ;p11 – Function describing transition between p9 & p10 (eg 3)

sr = 22050

kr = 1470

ksmps = 15


nchnls = 1

instr 1

```
p9 = p9*.01           ;1st am %
p10 = p10*.01         ;2nd am %
k1 linen p4, p6, p3, p7 ;amplitude envelope
k2 oscili p10-p9, 1/p3, p11 ;varying am
k2 = k1*k2            ;am level
k1 = k1 - k2          ;straight level
a1 oscili k2,cpspch(p5)*p8,4 ;am oscil
a1 oscili a1+k1,cpspch(p5),10 ;audio oscil
out a1
endin
```

**:CSTIL06.SC**

```
f3 0 129 7 0 64 1 64 0
f4 0 512 7 1 256 0 256 1
f10 0 1024 10 1 .7 .3 .6 .2 .5
i1    0      2.0    15000   8.00     .2      .5      1.487   20     80   3
i1    2      0.5    20000   7.05     .1      .3      1.113   40     80   3
e
```

**Flow Chart for CSTIL06:**

**;CSTIL07.ORC Linear and Exponential Glissando Instrument**

;Uses 11 parameters  
; p3 – Duration (eg 2)  
; p4 – Amplitude (eg 15000)  
; p5 – Pitch1 (eg Pch notation 8.00 for middle C)  
; p6 – Attack time of linear envelope (eg .2)  
; p7 – Decay time of linear envelope (eg .3)  
; p8 – Pitch2 (eg 7.00)  
; p9 – Duration 1st note is held before gliss occurs (eg .5)  
; p10 – Time taken over glissando  
; p11 – Glissando shape select (0 = linear, 1 = exponential)

sr = 22050

kr = 1470

ksmps = 15

nchnls = 1

instr 1

```
i1 = octpch(p5)          ;convert to oct.dec
i2 = octpch(p8)          ;ditto
k1 linen p4, p6, p3, p7  ;envelope
if p11 != 0 kgoto expgliss ;is exponential gliss required?
k2 linseg i1,p9,i1,p10,i2,p3-(p9+p10),i2 ;no,linear
if p11=0 kgoto doit      ;jump past expgloss

expgliss:
k2 expseg i1,p9,i1,p10,i2,p3-(p9+p10),i2 ;exponential

doit:
a1 oscili k1, cpsoct(k2), 7
```

out a1


endin

**;CSTIL07.SC**

f7 0 1024 10 1 .4 .5 .6 .3 .2

No	St.	Dur	Amp	Pch1	Att.	Dec.	Pch2	Hld	GI-T	Exp/Lin
;Glissando exponential										
i1	0	2	20000	8.00	.2	.5	7.00	.6	.6	1
;Glissando linear										
i1	2	2	20000	8.00	.2	.5	7.00	.6	.6	0
;Both together										
i1	4	3	12000	8.00	.2	.5	7.00	.6	.6	1
i1	4	3	12000	8.00	.2	.5	7.00	.6	.6	0

e

**Flow Chart for CSTIL07:**

**;CSTIL08.ORC Sub-Audio Frequency Modulation**  
 ;Uses 11 parameters  
 ;p3 – Duration (eg 2)  
 ;p4 – Amplitude (eg 15000)  
 ;p5 – Pitch (eg Pch notation 8.00 for middle C)  
 ;p6 – Attack time of linear envelope (eg .2)  
 ;p7 – Decay time of linear envelope (eg .3)  
 ;p8 – Vibrato Width (eg pch .015)  
 ;p9 – Vibrato Rate (eg 4.5 Hz)  
 ;p10 – Waveform select for timbre (eg 10 – three harmonics)  
 ;p11 – Waveform select for vibrato (eg 1 – a sine wave)

sr = 22050

kr = 1470

ksmps = 15

nchnls = 1

instr 1

i1 = octpch(p5)

;convert pch to octave.decimal

k1 linen p4, p6, p3, p7

;amplitude envelope

k2 oscili p8, p9, p11

;varying additional pitch (vibrato)

a1 oscili k1,cpsoct(i1+k2),p10

;audio osc. adding vibrato

out a1

endin

**;CSTIL08.SC**


f1 0 512 10 1

f10 0 1024 10 1 .1 .5

;PFields

1	2	3	4	5	6	7	8	9	10	11
i1	0	2.0	15000	8.00	.2	.5	0.015	4.5	10	1
e										

*Flow Chart for CSTIL08:*



**;CSTIL09.ORC Audio-rate Frequency Modulation**

;Sine modulating sine  
;Uses 11 parameters  
;p3 – Duration (eg 2)  
;p4 – Amplitude (eg 15000)  
;p5 – Pitch (eg Pch notation 8.00 for middle C)  
;p6 – Attack time of linear envelope (eg .2)  
;p7 – Decay time of linear envelope (eg .3)  
;p8 – Modulation multiplier (eg 1)  
;p9 – Modulation Index (eg 3)  
;p10 – Mod Rise-Time (eg .5)  
;p11 – Mod Fall-Time (eg .5)

sr = 22050

kr = 1470

ksmps = 15

nchnls = 1

instr 1

```
k1 linen p4, p6, p3, p7          ;envelope
k2 linen p9, p10, p3, p11        ;index
a1 oscili k1, cpspch(p5), 1, p8, k2, 10 ;fm using oscili
out a1
endin
```


**;CSTIL09.SC**

f10 0 1024 10 1

;PFields

1	2	3	4	5	6	7	8	9	10	11
No	St.	Dur	Amp	Pch	Att	Dec	Mult	Idx	Rise	Fall
i1	0	2	20000	8.00	.2	.5	1	3	.5	.5
i1	2	2	25000	8.07	.1	.7	.9	5	.5	.5
e										

*Flow Chart for CSTIL09:*



**:CSTIL10.ORC Buzz Generator with envelope and controllable number of partials**

;Uses 8 parameters  
 ;p3 – Duration (eg 2)  
 ;p4 – Amplitude (eg 15000)  
 ;p5 – Pitch (eg Pch notation 8.00 for middle C)  
 ;p6 – Attack time of linear envelope (eg .2)  
 ;p7 – Decay time of linear envelope (eg .3)  
 ;p8 – No. of partials in waveform (eg 12)

sr = 22050

kr = 1470

ksmps = 15

nchnls = 1

instr 1

k1 linen p4, p6, p3, p7  
 a1 buzz k1, cpspch(p5), p8, 1  
 out a1  
 endin


**:CSTIL10.SC**

f1 0 8193 10 1

;PFields

1 ;No.	2 St.	3 Dur	4 Amp	5 Pch	6 Att	7 Dec	8 No.Partial
i1	0	2	15000	8.00	.2	.5	12
i1	2	1	18000	8.06	.04	.2	8
i1	3	1	20000	8.01	.04	.2	10
i1	4	4	25000	7.07	.3	.8	14

e



*Flow Chart for CSTIL10*

**;CSTIL11.ORC Timbre-migrating instrument****;Uses 7 parameters****;p3 – Duration (eg 2)****;p4 – Amplitude (eg 15000)****;p5 – Pitch (eg Pch notation 8.00 for middle C)****;p6 – Attack time of linear envelope (eg .2)****;p7 – Decay time of linear envelope (eg .3)****;This instrument uses two simultaneous waveforms, and provides a transition between them.****;Linseg creates a transition control (k2) between the two, which is then used in the out statement.**

sr = 22050

kr = 1470

ksmps = 15

nchnls = 1

instr 1

k1 linen p4, p6, p3, p7 ;amplitude envelope

k2 linseg 1, p3/5, 1, p3\*3/5, 0, p3/5, 0 ;transition control

a1 oscili k1, cpspch(p5), 1 ;1st audio waveform

a2 oscili k1, cpspch(p5), 2 ;2nd audio waveform

out (a1\*k2)+(a2\*(1-k2))

endin

**;CSTIL11.SC**

f1 0 1024 10 1 .5 .3 ;Three Harmonics

f2 0 1024 10 81 0 27 0 9 0 3 0 1 0 .3 0 .1 ;Squarish


**;PFields**

1	2	3	4	5	6	7
No	St	Dur	Amp	Pch	Att	Dec


t 0 100 5 60

i1	0	6	15000	6.00	.5	.5
i1	0	1	12000	10.00	.1	.2
i1	1	1	12000	9.11	.1	.2
i1	2	1	12000	9.07	.1	.2
i1	3	1	12000	9.06	.1	.2
i1	4	2	12000	9.04	.2	.5

e

**Flow Chart for CSTIL11:**





Flow Chart for CSTIL12

